

『双対アプローチによる一般均衡モデルの記述』

武田史郎*

関東学園大学経済学部経済学科
373-8515 群馬県太田市藤阿久町 200

平成 20 年 6 月 4 日

\$Id: ge_duality_jp.tex,v 1.3 2008/06/04 12:50:06 st Exp \$

目次

1 導入	1
2 伝統的アプローチ	2
2.1 生産サイド	2
2.2 需要サイド	3
2.3 市場均衡	3
2.4 均衡条件	4
3 双対アプローチ	4
3.1 生産サイド	4
3.2 需要サイド	5
3.3 市場均衡	6
3.4 均衡条件	6
3.5 双対アプローチの特徴	6
4 例	7
4.1 伝統的アプローチ	7
4.2 双対アプローチ	8
A GAMS のプログラム	9

1 導入

一般均衡モデルを記述する方法として「双対アプローチ (dual approach)」と呼ばれるものが使われることがある。この文脈での「双対アプローチ」とはモデル (均衡条件) を記述する際に、生産関数、効用関数、非補償需要関数 (マーシャルの需要関数) 等を使わず、費用関数、支出関数、条件付き要素需要関数、補償需要関数 (ヒックスの需要関数) 等を用いる方法を指している。本稿では、

*email: <shiro.takeda@gmail.com>

この双対アプローチについて、伝統的なアプローチと比較しながら説明をおこなう。以下では、簡単なモデルを例にとり、まず伝統的アプローチを見た後に、双対アプローチを説明することにする。また、appendix に同じモデルを伝統的なアプローチと双対アプローチで記述しモデルを解くための GAMS のプログラムを付けている。

2 伝統的アプローチ

まず、生産関数、効用関数、非補償需要関数等による伝統的なアプローチを用いてモデルを記述する。例とするモデルは静学的な完全競争モデルである。二つの消費財 $i = 1, 2$ を生産する二つの部門が存在し、資本・労働を用いて規模に関して収穫一定の技術のもとで生産をおこなうとする。需要サイドとしては、一つの代表的消費者(家計)が存在し、生産活動に生産要素を提供するとともに、それにより得た収入をもとに消費財を購入・消費しているものとする。家計の保有する生産要素の賦存在量は一定であるとする。モデルは貿易のない閉鎖経済とし、さらに政府部門は存在しないものとする。

2.1 生産サイド

生産関数は次式で与えられるものとする。

$$Q_i = f_i(K_i, L_i) \quad (1)$$

Q_i は財 i の生産量、 K_i 、 L_i はそれぞれ財 i の生産における資本、労働の投入量である。収穫一定の技術であるので、生産関数 $f_i(\cdot)$ は一次同次性を満たす。

p_i を財 i の価格、 r 、 w をそれぞれレンタルプライス、賃金率とすると、財 i を生産する部門 i の利潤 π_i は以下のように表現できる。

$$\pi_i = p_i f_i(K_i, L_i) - rK_i - wL_i$$

生産者はプライステイカーであるので、利潤最大化条件は以下で与えられる。

$$\begin{aligned} \frac{\partial \pi_i}{\partial K_i} = 0 &: p_i \frac{\partial f_i(K_i, L_i)}{\partial K_i} = r \\ \frac{\partial \pi_i}{\partial L_i} = 0 &: p_i \frac{\partial f_i(K_i, L_i)}{\partial L_i} = w \end{aligned}$$

これは限界生産物価値(左辺)を要素価格(右辺)に均等化させるという意味である。利潤最大化を目指す生産者は、財の価格 p_i 、要素価格 r 、 w を所与とした上で、この条件を満たすように生産要素の投入量 K_i 、 L_i を決定する。最適な生産要素の投入量が決めれば、(1) より最適な供給量が決まることになる¹。

¹厳密には、生産関数の一次同次性により要素投入量の絶対的な水準は決まらず、投入比率しか決まらない。従って、最適な生産量の絶対的な水準は不決定と(無数に存在すること)になる。ただし、需要サイドも考慮すれば均衡における生産量は結局一意に決まることになるので、生産サイドの不決定性は問題にはならない。

2.2 需要サイド

消費者(家計)は二つの財を消費することからのみ効用を得るので、財 i の消費量を C_i とすると、効用関数は次式で与えられる。

$$U = u(C_1, C_2)$$

本稿ではこの効用関数が一次同次であると仮定する。序数的な観点からのみ効用を測るとするならば、一次同次性を仮定する必要はなく、homothetic という仮定で十分である。しかし、ここではシミュレーション(特に、CGE 分析)への応用を念頭に置いているので、シミュレーションで通常仮定されるように一次同次であるとする²。理論分析では序数的な観点にとられることが普通であるので、一次同次性が仮定されることはあまりないが、シミュレーションではよく置かれる仮定である。

M を消費者の所得とすると、消費者の効用最大化問題は次のように定式化される。

$$\begin{aligned} \max u(C_1, C_2) \\ \text{s.t. } p_1 C_1 + p_2 C_2 \leq M \end{aligned}$$

この効用最大化問題に対する Lagrange 関数は次式で与えられる。

$$\mathcal{L} = u(C_1, C_2) - \lambda [p_1 C_1 + p_2 C_2 - M]$$

効用最大化条件は以下で与えられる。

$$\begin{aligned} \frac{\partial \mathcal{L}}{\partial C_i} = 0: \quad \frac{\partial u(C_1, C_2)}{\partial C_i} = \lambda p_i \quad i = 1, 2 \\ \frac{\partial \mathcal{L}}{\partial \lambda} = 0: \quad p_1 C_1 + p_2 C_2 = M \end{aligned}$$

この3本の条件より、非補償需要関数 $C_i^M(p_1, p_2, M)$ が導出される。

所得は一定量を所有する生産要素からの要素所得である。資本の賦存量、労働の賦存量をそれぞれ K 、 L で表すとすると、消費者の所得 M は次式で与えられる。

$$M = rK + wL$$

2.3 市場均衡

ここまで生産者、消費者の行動を定式化した。そこから導かれる供給、需要をもとに市場均衡を求める。市場としては、財市場と生産要素市場がある。

財 i の供給量は Q_i 、財 i への需要量は C_i^M で与えられた。よって、財市場の均衡は次式となる。

$$Q_i = C_i^M(p_1, p_2, M)$$

資本・労働は消費者によって K 、 L だけ供給され、生産者によって K_i 、 L_i だけ需要される。よって要素市場の均衡条件は次式となる。

$$K = \sum_i K_i \quad L = \sum_i L_i$$

²CGE では厚生の変化を観察する際に等価変分が利用されることが多いが、等価変分は効用関数が homothetic の場合と、さらに強い仮定である一次同次の場合とで全く同じ値になるので、どちらを仮定しておいても変わらない。しかし、実質所得の変化率(「等価変分/初期の所得」)を計算する際には、効用関数を一次同時としておくことで計算が簡単となるという利点がある。

2.4 均衡条件

以上で導出した均衡条件をまとめる。

$$Q_i = f_i(K_i, L_i) \quad \{Q_i\}_{i=1,2} \quad (2)$$

$$p_i \frac{\partial f_i(K_i, L_i)}{\partial K_i} = r \quad \{K_i\}_{i=1,2} \quad (3)$$

$$p_i \frac{\partial f_i(K_i, L_i)}{\partial L_i} = w \quad \{L_i\}_{i=1,2} \quad (4)$$

$$M = rK + wL \quad \{M\} \quad (5)$$

$$K = \sum_i K_i \quad \{K\} \quad (6)$$

$$L = \sum_i L_i \quad \{L\} \quad (7)$$

$$Q_i = C_i^M(p_1, p_2, M) \quad \{p_i\}_{i=1,2} \quad (8)$$

(2) は供給量、(3)–(4) は利潤最大化条件、(5) は所得の定義式、(6)–(7) は要素市場の均衡条件、(8) は財市場の均衡条件である。{ } 内の変数は、各条件がどの変数に対応するかを表している。これらの 11 本の均衡条件によって、生産量 $\{Q_i\}_i$ 、要素投入量 $\{K_i\}_i$ 、 $\{L_i\}_i$ 、所得 M 、要素価格 r 、 w 、財価格 $\{p_i\}_i$ の 11 個の変数が決定されることになる。ただし、ワルラスの法則より市場均衡条件のうち一本は **redundant** であるので、実際に解く際には一つの財・要素をニューメレールとし、それに対応する市場均衡条件を一本除いてやればよい³。

3 双対アプローチ

前節のモデルを双対アプローチを用いて記述する。

3.1 生産サイド

双対性より、利潤最大化を図る生産者は費用最小化を図る⁴。よって、費用関数を定義することができるが、生産関数 (1) は一次同次と仮定されていたので、生産量一単位あたりの単位費用関数 $c_i(r, w)$ を定義することができる⁵。

$$c_i(r, w) \equiv \min [rK + wL | f_i(K, L) = 1]$$

この単位費用関数を使うと利潤を次式で表現できる。

$$\pi_i = [p_i - c_i(r, w)] Q_i$$

³ワルラス法則とは、全ての市場の超過需要額の和はゼロとなる性質である。ワルラス法則が成り立つような状況では、 n 個の市場のうち $n - 1$ 個で均衡が成立していれば、残りの一つの市場でも必ず均衡が成立することになる。本稿のモデルでもワルラス法則が成り立つので、市場均衡条件のうち一本は余分 (**redundant**) となる。ワルラスの法則 (Walras' law) については、例えば奥野・鈴木 (1988, p. 17)、Mas-Colell, Whinston and Green (1995, p. 582) を参照。

⁴生産サイドにおける双対性 (費用関数、条件付き要素需要関数、Shepard's lemma) については、奥野・鈴木 (1985)、Mas-Colell et al. (1995, Chap. 5) を参照。

⁵生産関数が一次同次であるとき、費用関数が単位費用と生産量に分離できることは以下のように示せる。まず、 $\tilde{c}(w, r, Q)$ で通常の費用関数、つまり w と r を所与としたとき生産量 Q を実現するのに必要な最小費用を表すとす。 $\tilde{c}(w, r, Q)$ は数式では次式のように定義される。

$$\tilde{c}(w, r, Q) \equiv \min [rK + wL | f(K, L) = Q]$$

これより、利潤最大化条件は次式となる⁶。

$$\frac{\partial \pi_i}{\partial Q_i} = 0: p_i = c_i(r, w)$$

さらに Shephard の補題より、単位費用関数から単位要素需要関数を導出できる。

$$a_i^K = \frac{\partial c_i(r, w)}{\partial r} \quad a_i^L = \frac{\partial c_i(r, w)}{\partial w}$$

3.2 需要サイド

伝統的アプローチでは消費者が効用最大化行動をとるという仮定に基づいて需要サイドを扱ったが、双対性より効用最大化行動をとる消費者は支出最小化を図っていることになるので、消費の行動は支出関数でも表現できる⁷。効用関数を一次同時と仮定したので、支出関数に関しても単位支出関数を定義できる⁸。

$$e(p_1, p_2) \equiv \min [p_1 C_1 + p_2 C_2 | u(C_1, C_2) = 1]$$

定義から明かなように、 $e(p_1, p_2)$ は一単位の効用を得るのに必要な最小支出を表す関数、つまり単位支出関数である。

Shepard の補題より、この単位支出関数から単位補償需要 (効用一単位を得るのに必要な消費量) を導出できる。

$$a_i^C = \frac{\partial e(p_1, p_2)}{\partial p_i}$$

総補償需要 (ヒックスの需要関数) は、単位補償需要に効用をかけあわせた

$$a_i^C U$$

で与えられる。

双対性より、消費者が効用最大化を図っているならば同時に最小化された支出が所得に等しくなっていなければならない⁹。よって、以下の関係が成立する。

$$\tilde{e}(p_1, p_2, U) = e(p_1, p_2)U = M$$

これを f の一次同次性をもとに書き換えていく。

$$\begin{aligned} \tilde{c}(w, r, Q) &\equiv \min [rK + wL | f(K, L) = Q] \\ &= \min [rK + wL | f(K/Q, L/Q) = 1] \\ &= \min [r\tilde{K}Q + w\tilde{L}Q | f(\tilde{K}, \tilde{L}) = 1] \\ &= Q \min [r\tilde{K} + w\tilde{L} | f(\tilde{K}, \tilde{L}) = 1] \\ &= Q\tilde{c}(w, r, 1) \end{aligned}$$

$\tilde{c}(w, r, 1)$ は単位費用を表す。これが本文内の $c(r, w)$ である。

⁶注 1 と同様に、ここでも生産量が不決定となる問題がある。

⁷生産サイドにおける双対性 (支出関数、補償需要関数) については、奥野・鈴木 (1985)、Mas-Colell et al. (1995, Chap. 3) を参照。

⁸これは生産関数が一次同次の場合に単位費用関数を定義できるのと同じ理由による。

⁹例えば、奥野・鈴木 (1985, p. 193)、Mas-Colell et al. (1995, p. 60) を参照。

3.3 市場均衡

再び市場均衡条件を導出しよう。伝統的アプローチと同様に供給量は Q_i で表されるが、需要量は非補償需要関数ではなく、補償需要関数で表される。よって、財市場の均衡条件は次式となる。

$$Q_i = a_i^C U$$

要素に対する需要も単位需要関数によって表される。

$$K = \sum_i a_i^K Q_i \quad L = \sum_i a_i^L Q_i$$

3.4 均衡条件

双対アプローチによる均衡条件をまとめよう。

$$c_i(r, w) = p_i \quad \{Q_i\}_{i=1,2} \quad (9)$$

$$a_i^K = \frac{\partial c_i(r, w)}{\partial r} \quad \{a_i^K\}_{i=1,2} \quad (10)$$

$$a_i^L = \frac{\partial c_i(r, w)}{\partial w} \quad \{a_i^L\}_{i=1,2} \quad (11)$$

$$e(p_1, p_2)U = M \quad \{U\} \quad (12)$$

$$a_i^C = \frac{\partial e(p_1, p_2)}{\partial p_i} \quad \{a_i^C\}_{i=1,2} \quad (13)$$

$$M = rK + wL \quad \{M\} \quad (14)$$

$$Q_i = a_i^C U \quad \{p_i\}_{i=1,2} \quad (15)$$

$$K = \sum_i a_i^K Q_i \quad \{r\} \quad (16)$$

$$L = \sum_i a_i^L Q_i \quad \{w\} \quad (17)$$

(9) が利潤最大化条件、(10)–(11) は単位要素需要、(12) は効用最大化条件、(13) は単位補償需要、(14) は所得の定義式、(15) は財市場の均衡条件、(16)–(17) は要素市場の均衡条件である。この 14 本の条件から、生産量 Q_i 、単位要素需要 a_i^K 、 a_i^L 、効用水準 U 、単位補償需要 a_i^C 、所得水準 M 、財価格 p_i 、要素価格 r 、 w の 14 個の変数が決まる。

3.5 双対アプローチの特徴

(2)–(8) までの伝統的なアプローチによる均衡条件と (9)–(17) までの双対アプローチによる均衡条件を比較して欲しい。二つのアプローチの最も大きな違いは、伝統的アプローチでは生産サイドと需要サイドの記述の形式が異なっているのに対し、双対アプローチでは両者を同じような形式で記述できるという点である。

伝統的なアプローチでは、生産サイドは生産関数、「限界生産物価値＝要素価格」という利潤最大化条件、需要サイドは非補償需要関数によって表現されている。このため生産サイドと需要サイドで記述の形式が異なる。これに対し、双対アプローチでは生産サイドと需要サイドがほぼ同じ形式で記述されている。生産サイドと需要サイドがどう対応しているかは表 1 を見てほしい。つまり、

表 1: 双対アプローチにおける生産サイドと需要サイドの対応

条件	生産サイド	需要サイド
利潤最大化・効用最大化	$c_i(r, w) = p_i$	$e(p_1, p_2) = M/U$
単位需要	$a_i^K = \frac{\partial c_i(r, w)}{\partial r}$ $a_i^L = \frac{\partial c_i(r, w)}{\partial w}$	$a_i^C = \frac{\partial e(p_1, p_2)}{\partial p_i}$

単位費用関数と単位支出関数、条件付き要素需要関数と補償需要関数というように生産サイドと需要サイドがきれいに対応しているのである。

この形式上の対称性はシミュレーションをおこなう際に大きなメリットとなる。シミュレーションをおこなうには、生産関数・効用関数の関数型を具体的に特定化した上で、そこから様々な関数を導出する必要がある。ここで仮に生産関数と効用関数が同じ関数型 (例えば、CES 型) であったとしても、伝統的アプローチでは、生産サイドは限界生産物、需要サイドは非補償需要という全く形式の異なったものを求める必要がある。一方、双対アプローチでは、生産サイドの単位費用関数と需要サイドの単位支出関数、生産サイドの条件付き需要と需要サイドの補償需要が全く同じ形式であるので、どちらかがわかればもう片方を計算する必要はなくなる。よって、関数の導出という手間を省くことができるのである。

貿易理論における一般均衡分析では、この双対アプローチが使われることが多い。貿易理論における双対アプローチについては、例えば、Dixit and Norman (1980)、Wong (1995)、小田 (1997) 等を参照してほしい¹⁰。

4 例

ここでは例として生産関数、効用関数を特定化し、均衡条件を記述する。生産関数、効用関数はそれぞれ以下のように Cobb-Douglas 型と仮定する。

$$f_i(K_i, L_i) = \phi_i(K_i)^{\alpha_i}(L_i)^{1-\alpha_i}$$

$$u(C_1, C_2) = \psi(C_1)^{\beta_1}(C_2)^{\beta_2} \text{ where } \beta_1 + \beta_2 = 1$$

4.1 伝統的アプローチ

限界生産物：

$$\frac{\partial f_i(K_i, L_i)}{\partial K_i} = \phi_i \alpha_i \left[\frac{L_i}{K_i} \right]^{1-\alpha_i}$$

$$\frac{\partial f_i(K_i, L_i)}{\partial L_i} = \phi_i (1 - \alpha_i) \left[\frac{K_i}{L_i} \right]^{\alpha_i}$$

¹⁰ただし、貿易理論では生産サイドを表現するのに費用関数ではなく、収入関数 (GDP 関数) と呼ばれるものを利用していることが多い。

非補償需要関数：

$$C_i^M = \frac{\beta_i M}{p_i}$$

均衡条件：

$$\begin{aligned} Q_i &= \phi_i (K_i)^{\alpha_i} (L_i)^{1-\alpha_i} \\ p_i \phi_i \alpha_i \left[\frac{L_i}{K_i} \right]^{1-\alpha_i} &= r \\ p_i \phi_i (1 - \alpha_i) \left[\frac{K_i}{L_i} \right]^{\alpha_i} &= w \\ M &= rK + wL \\ K &= \sum_i K_i \\ L &= \sum_i L_i \\ Q_i &= \frac{\beta_i M}{p_i} \end{aligned}$$

4.2 双対アプローチ

単位費用関数：

$$c_i(r, w) = \frac{1}{\phi_i} \left[\frac{r}{\alpha_i} \right]^{\alpha_i} \left[\frac{w}{1 - \alpha_i} \right]^{1-\alpha_i}$$

単位要素需要関数：

$$\begin{aligned} a_i^K &= \frac{\partial c_i(r, w)}{\partial r} = \frac{\alpha_i}{r} \frac{1}{\phi_i} \left[\frac{r}{\alpha_i} \right]^{\alpha_i} \left[\frac{w}{1 - \alpha_i} \right]^{1-\alpha_i} = \frac{\alpha_i c_i(r, w)}{r} \\ a_i^L &= \frac{\partial c_i(r, w)}{\partial w} = \frac{(1 - \alpha_i)}{w} \frac{1}{\phi_i} \left[\frac{r}{\alpha_i} \right]^{\alpha_i} \left[\frac{w}{1 - \alpha_i} \right]^{1-\alpha_i} = \frac{(1 - \alpha_i) c_i(r, w)}{w} \end{aligned}$$

単位支出関数：

$$e(p_1, p_2) = \frac{1}{\psi} \left[\frac{p_1}{\beta_1} \right]^{\beta_1} \left[\frac{p_2}{\beta_2} \right]^{\beta_2}$$

単位補償需要関数：

$$a_i^C = \frac{\partial e(p_1, p_2)}{\partial p_i} = \frac{\beta_i}{p_i} \frac{1}{\psi} \left[\frac{p_1}{\beta_1} \right]^{\beta_1} \left[\frac{p_2}{\beta_2} \right]^{\beta_2} = \frac{\beta_i e(p_1, p_2)}{p_i}$$

均衡条件：

$$c_i = \frac{1}{\phi_i} \left[\frac{r}{\alpha_i} \right]^{\alpha_i} \left[\frac{w}{1 - \alpha_i} \right]^{1 - \alpha_i}$$

$$c_i = p_i$$

$$a_i^K = \frac{\alpha_i c_i}{r}$$

$$a_i^L = \frac{(1 - \alpha_i) c_i}{w}$$

$$e = \frac{1}{\psi} \left[\frac{p_1}{\beta_1} \right]^{\beta_1} \left[\frac{p_2}{\beta_2} \right]^{\beta_2}$$

$$eU = M$$

$$a_i^C = \frac{\beta_i e}{p_i}$$

$$M = rK + wL$$

$$Q_i = a_i^C U$$

$$K = \sum_i a_i^K Q_i$$

$$L = \sum_i a_i^L Q_i$$

参考文献

Dixit, Avinash K. and Victor Norman (1980) *Theory of International Trade: A Dual General Equilibrium Approach*, New York: Cambridge University Press.

Mas-Colell, Andreu, Michael D. Whinston, and Jerry R. Green (1995) *Microeconomic Theory*, New York: Oxford University Press.

Wong, Kar-yiu (1995) *International Trade in Goods and Factor Mobility*, Cambridge, MA: MIT Press.

奥野正寛・鈴木興太郎 (1985) 『ミクロ経済学 I』, 岩波書店. モダン・エコノミクス 1.

—— (1988) 『ミクロ経済学 II』, 岩波書店. モダン・エコノミクス 2.

小田正雄 (1997) 『現代国際経済学』, 有斐閣.

A GAMS のプログラム

同じモデルを二つのアプローチで解く GAMS のプログラム。

```
$title A sample GAMS program.
display "com: A sample GAMS program.";
$ontext
Copyright (C) 2004 Shiro Takeda
Time-stamp: <2004-10-27 12:14:22 Shiro Takeda>
version: 1.1
```

\$Id: ge_duality.gms,v 1.1.1.1 2006/09/26 15:17:16 st Exp \$
 Author: Shiro Takeda <zbc08106@park.zero.ad.jp>
 e-mail: <zbc08106@park.zero.ad.jp>
 First-written: <2004/10/26>
 \$offtext

```

* -----
* Set definitions:
set      i          Index of goods and sectors      / 1, 2 /;
display i;

* -----
* Parameter definitions:
parameter
  phi(i)          Scale parameter in production functions
  psi             Scale parameter in utility function
  alpha(i)        Share of capital in production functions
  beta(i)         Share parameter in utility function
;
alpha("1") = 0.4;
alpha("2") = 0.6;
beta("1") = 2/3;
beta("2") = 1 - beta("1");
display alpha, beta;

phi("1") = 200 / ((80)**(alpha("1")) * (120)**(1-alpha("1")));
phi("2") = 100 / ((60)**(alpha("2")) * (40)**(1-alpha("2")));
display phi;

psi = 300 / ((200)**(beta("1")) * (100)**(beta("2")));
display psi;

* -----
* Declare exogenous variables
parameter
  end_l          Labor endowment          / 160 /
  end_k          Capital endowment        / 180 /;
display end_l, end_k;

* -----
* Declare variables
variables
* Variables for traditional approach
  q(i)          Output
  k(i)          Capital input
  l(i)          Labor input
  p(i)          Goods prices
  r            Rental price
  w            Wage rate
  m            Income

* Variables for dual approach
  q_(i)         Output
  c_(i)         Unit cost
  a_k_(i)       Unit demand for capital
  a_l_(i)       Unit demand for labor
  u_           Utility
  e_           Unit expenditure

```

```

a_c_(i)    Unit consumption demand
m_         Income
p_(i)     Goods prices
r_         Rental price
w_         Wage rate
;
*
* -----
*   Declare equations
equations
*   Equations for traditional approach
e_q(i)     Outpu
e_k(i)     Capital input
e_l(i)     Labor input
e_p(i)     Goods price
e_r        Rental price
e_w        Wage rate
e_m        Income

*   Equations for dual approach
e_q_(i)    Output
e_c_(i)    Unit cost
e_a_k_(i)  Unit demand for capital
e_a_l_(i)  Unit demand for labor
e_u_       Utility
e_e_       Unit expenditure
e_a_c_(i)  Unit consumption demand
e_m_       Income
e_p_(i)    Goods prices
e_r_       Rental price
e_w_       Wage rate
;
*
* -----
*   Definitions of equations for traditional approach:

*   Production function:
e_q(i) ..

    phi(i) * (k(i))**(alpha(i)) * (l(i))**(1-alpha(i)) =e= q(i);

*   FOC of profit maximization w.r.t captial:
e_k(i) ..

    r =e= p(i) * phi(i) * alpha(i) * (l(i)/k(i))**(1-alpha(i));

*   FOC of profit maximization w.r.t labor:
e_l(i) ..

    w =e= p(i) * phi(i) * (1-alpha(i)) * (k(i)/l(i))**(alpha(i));

*   Income:
e_m ..

    r * end_k + w * end_l =e= m;

*   Market for capital:
e_r ..

```

```

    end_k =e= sum(i, k(i));

*      Market for labor:
e_w ..

    end_l =e= sum(i, l(i));

*      Market for goods:
e_p(i) ..

    q(i) =e= beta(i) * m / p(i);

*      -----
*      Model declaration for traditional approach:
model ge_trad Traditional approach /
    e_q.q, e_k.k, e_l.l, e_m.m, e_r.r, e_w.w, e_p.p /;
*      -----
*      Definitions of equations for dual approach:

*      FOC of profit maximization:
e_q_(i) ..

    c_(i) =e= p_(i);

*      Unit cost function:
e_c_(i) ..

    c_(i) =e= (1/phi(i))
    * (r_ / alpha(i))**(alpha(i)) * (w_ / (1-alpha(i)))**(1-alpha(i));

*      Unit demand for capital:
e_a_k_(i) ..

    a_k_(i) =e= alpha(i) * c_(i) / r_;

*      Unit demand for labor:
e_a_l_(i) ..

    a_l_(i) =e= (1-alpha(i)) * c_(i) / w_;

*      Utility maximization:
e_u_ ..

    e_ * u_ =e= m_;

*      Unit expenditure function:
e_e_ ..

    e_ =e= (1/psi) * prod(i, (p_(i)/beta(i))**(beta(i)));

*      Unit consumption demand:
e_a_c_(i) ..

    a_c_(i) =e= beta(i) * e_ / p_(i);

*      Income:
e_m_ ..

```

```

    r_ * end_k + w_ * end_l =e= m_;

*      Market for goods:
e_p_(i) ..

    q_(i) =e= a_c_(i) * u_;

*      Market for capital:
e_r_ ..

    end_k =e= sum(i, a_k_(i) * q_(i));

*      Market for labor:
e_w_ ..

    end_l =e= sum(i, a_l_(i) * q_(i));

*      -----
*      Model declaration for dual approach:
model ge_dual Dual approach /
    e_q_.q_, e_c_.c_, e_a_k_.a_k_, e_a_l_.a_l_, e_u_.u_, e_e_.e_,
    e_a_c_.a_c_, e_m_.m_, e_p_.p_, e_r_.r_, e_w_.w_ /;

*      -----
*      Lower bound for variables:
q_.lo(i) = 1e-6;
k_.lo(i) = 1e-6;
l_.lo(i) = 1e-6;
p_.lo(i) = 1e-6;
r_.lo = 1e-6;
w_.lo = 1e-6;
m_.lo = 1e-6;

q_.lo(i) = 1e-6;
c_.lo(i) = 1e-6;
a_k_.lo(i) = 1e-6;
a_l_.lo(i) = 1e-6;
u_.lo = 1e-6;
e_.lo = 1e-6;
a_c_.lo(i) = 1e-6;
m_.lo = 1e-6;
p_.lo(i) = 1e-6;
r_.lo = 1e-6;
w_.lo = 1e-6;

*      -----
*      Initial variables for variables:
q.l("1") = 200;
q.l("2") = 100;
k.l("1") = 80;
k.l("2") = 60;
l.l("1") = 120;
l.l("2") = 40;
l.l(i) = 1;
p.l(i) = 1;
r.l = 1;
w.l = 1;

```

```

m.l = 300;

q_.l("1") = 200;
q_.l("2") = 100;
c_.l(i) = 1;
a_k_.l("1") = 80/200;
a_k_.l("2") = 60/100;
a_l_.l("1") = 120/200;
a_l_.l("2") = 40/100;
u_.l = 300;
e_.l = 1;
a_c_.l("1") = 200/300;
a_c_.l("2") = 100/300;
m_.l = 300;
p_.l(i) = 1;
r_.l = 1;
w_.l = 1;

* -----
* Set numeraire goods (labor):
w.fx = 1;
w_.fx = 1;

* -----
* Solve two models:
display "com: Solve two models:"

parameter
    res          Compare results from two alternative models;

* Traditional approach:
solve ge_trad using mcp;

res("q1","trad") = q_.l("1");
res("q2","trad") = q_.l("2");
res("c1","trad") = beta("1") * m.l / p.l("1");
res("c2","trad") = beta("2") * m.l / p.l("2");
res("util","trad")
    = psi * (res("c1","trad"))**(beta("1")) * (res("c2","trad"))**(beta("2"));

* Dual approach:
solve ge_dual using mcp;

res("q1","dual") = q_.l("1");
res("q2","dual") = q_.l("2");
res("c1","dual") = a_c_.l("1") * u_.l;
res("c2","dual") = a_c_.l("2") * u_.l;
res("util","dual") = u_.l;
display res;

* -----
* Changes in factor endowments:
display "com: Changes in factor endowments:";

end_l = 1.2 * end_l;
end_k = 0.9 * end_k;
display end_l, end_k;

```

```

*      Traditional approach:
solve ge_trad using mcp;

res("q1","trad") = q.l("1");
res("q2","trad") = q.l("2");
res("c1","trad") = beta("1") * m.l / p.l("1");
res("c2","trad") = beta("2") * m.l / p.l("2");
res("util","trad")
    = psi * (res("c1","trad"))**(beta("1")) * (res("c2","trad"))**(beta("2"));

*      Dual approach:
solve ge_dual using mcp;

res("q1","dual") = q_.l("1");
res("q2","dual") = q_.l("2");
res("c1","dual") = a_c_.l("1") * u_.l;
res("c2","dual") = a_c_.l("2") * u_.l;
res("util","dual") = u_.l;
display res;

* -----
* Local Variables:
* mode: gams
* fill-column: 78
* End:

```